EvaluateFormIntegrals

`EvaluateFormIntegrals[`*expr*`]`
will evaluate any `FormIntegrals` in expression using *Mathematica* `Integrate` .

Details

- `EvaluateFormIntegrals` will accept `FormIntegral` domains that are either in the form of inequalities or in the form of *Mathematica* geometric Regions. The routine automatically generates the proper form of Integrate assuming that what is not a Region will be a proper inequality specification.
- For non-Cartesian coordinate systems with inequalities the form should include a volume factor. With spherical geometric Regions such as `Ball` or `Sphere` *Mathematica* automatically inserts the volume factor so it should be left out of the form expression.
- `EvaluateFormIntegrals` accepts `Integrate` options. The most common one would be specifying `Assumptions` for any parameters in the integral. In addition, any parameters should be declared as Grassmann scalars.
- An extra option is InactiveIntegral with the default value of False. When set to True, the Integrate statement will be in the Inactive form and Activate must be used to complete the evaluation. This allows inspection of the generated Integrate statement.
- The integrations are always performed in the direction of increasing coordinate variables, which defines the positive orientation.
- The GeometryPalette, which is included in the GrassmannCalculus palettes is useful when using Region domain specifications.

Examples (10)

Basic Examples (1)

In[1]:=
```
<< GrassmannCalculus`
```

In[2]:=
```
SetEuclideanNSpace[2, {x, y}, "Form"]
```

To find the area between two curves;

In[3]:=
```
domain = π / 4 ≤ x ≤ 5 π / 4 && Cos[x] ≤ y ≤ Sin[x];
FormIntegral[domain, dx∧dy]
EvaluateFormIntegrals[%]
```

Out[3]= $\int_{\mathcal{Domain}} dx \wedge dy$

Out[3]= $2\sqrt{2}$

©1988–2015 Wolfram Research, Inc. All rights reserved. http://reference.wolfram.com/language

Show the unevaluated *Mathematica* Integrate statement.

In[4]:=
```
domain = π / 4 ≤ x ≤ 5 π / 4 && Cos[x] ≤ y ≤ Sin[x];
FormIntegral[domain, dx∧dy]
EvaluateFormIntegrals[%, InactiveIntegral → True]
Activate[%]
```

Out[4]= $\int_{Domain} dx \wedge dy$

Out[4]= $\int_{-\infty}^{\infty}\int_{-\infty}^{\infty} \text{Boole}\left[\frac{\pi}{4} \le x \le \frac{5\pi}{4} \,\&\&\, \text{Cos}[x] \le y \le \text{Sin}[x]\right] dy\, dx$

Out[4]= $2\sqrt{2}$

Evaluate an integral that uses a Region and a parameter. Without an Assumptions option we obtain a `ConditionalExpression`. We also show the unevaluated `Integrate` statement generated.

In[5]:=
```
★★S[R]
```

In[6]:=
```
FormIntegral[Ball[{0, 0}, R], dx∧dy]
EvaluateFormIntegrals[%, InactiveIntegral → True]
Activate[%]
```

Out[6]= $\int_{Domain} dx \wedge dy$

Out[6]= $\int_{\{x,y\}\in \text{Ball}[\{0,0\},R]} 1$

Out[6]= $\text{ConditionalExpression}\left[\pi R^2, R > 0\right]$

In the following we add an Assumption.

In[7]:=
```
FormIntegral[Ball[{0, 0}, R], dx∧dy]
EvaluateFormIntegrals[%, Assumptions → R > 0]
```

Out[7]= $\int_{Domain} dx \wedge dy$

Out[7]= πR^2

Scope (9)

1-Dimensional Integrals (1)

Line Integrals (5)

With line integrals we will have a function that maps a 1-dimensional interval to a higher k-dimensional space. We evaluate the integral by pulling back the form to the 1-dimensional space.

Circumference of a Circle (1)

Length of One Turn of a Helix (1)

Length of a Curve Specified by an Equation (1)

Sometimes we can evaluate an arbitrary 1-form in a higher dimensional space by using equations for all but one of the variables.

In[11]:=
```
SetEuclideanNSpace[2, {x, y}, "Form"]
```

©1988–2015 Wolfram Research, Inc. All rights reserved. http://reference.wolfram.com/language

Evaluate the 1-form:

In[12]:=
```
oneform := y dx - x^2 dy
```

On the curve given by:

In[13]:=
```
eqn := y == 4 x - x^2
yRule = eqn /. Equal → Rule
```

Out[14]= $y \to 4x - x^2$

Write the form with an exterior derivative for dy:

In[15]:=
```
oneform /. dy -> d[y]
% /. yRule
newform = % // EvaluateExteriorDerivatives
```

Out[15]= $dx\, y - x^2\, \mathbf{d}[y]$

Out[16]= $dx\left(4x - x^2\right) - x^2\, \mathbf{d}\left[4x - x^2\right]$

Out[17]= $dx\left(4x - 5x^2 + 2x^3\right)$

That gives us a 1-dimensional integral. Define the domain as $0 \le x \le 4$.

In[18]:=
```
SetEuclideanNSpace[1, {x}, "Form"]
FormIntegral[0 ≤ x ≤ 4, newform]
% // EvaluateFormIntegrals
```

Out[19]= $\int_{Domain} dx\left(4x - 5x^2 + 2x^3\right)$

Out[20]= $\frac{160}{3}$

Work Against a Gravitational Field (1)

2-Dimensional Integrals (3)

Area of a Circle (1)

Surface Area of a Sphere (1)

Integrating a 2-Form on a Plane (1)

In[116]:=
```
SetEuclideanNSpace[3, {x, y, z}, "Vector"]
```

Evaluate the two form $(y^2 + 2yz)\, dS$ on the portion of the plane $2x + y + 2z == 6$ on the portion that's in the postive octant. The intersections of the plane with the three axes are:

In[98]:=
```
{ptx, pty, ptz} = #.Basis & /@ {{3, 0, 0}, {0, 6, 0}, {0, 0, 3}}
```

Out[98]= $\{3\, e_x,\ 6\, e_y,\ 3\, e_z\}$

The domain is:

In[100]:=
```
Solve[2 x + y == 6, y][[1, 1]] // FullSimplify
```

Out[100]= $y \to 6 - 2x$

In[101]:=
```
xyDomain = 0 ≤ x ≤ 3 && 0 ≤ y ≤ 6 - 2 x;
```

©1988–2015 Wolfram Research, Inc. All rights reserved. http://reference.wolfram.com/language

We will need a pullback for z.

In[102]:=
```
zPullback = Solve[2 x + y + 2 z == 6, z][[1, 1]]
```

Out[102]= $z \to \frac{1}{2}(6 - 2x - y)$

The tangent plane and its pullback are:

In[117]:=
```
step1 = GrassmannNormalize[(ptx - ptz) ∧ (pty - ptz)]
step2 = step1 /. VectorToForm
step3 = step2 /. dz → d[z] /. zPullback
step4 = step3 // EvaluateExteriorDerivatives
xyForm = (y^2 + 2 y z) step4 /. zPullback // Simplify
```

Out[117]= $\frac{1}{27}(3 e_x - 3 e_z) \wedge (6 e_y - 3 e_z)$

Out[118]= $\frac{1}{27}(3\,dx - 3\,dz) \wedge (6\,dy - 3\,dz)$

Out[119]= $\frac{1}{27}\left(3\,dx - 3\,\mathbf{d}\left[\frac{1}{2}(6 - 2x - y)\right]\right) \wedge \left(6\,dy - 3\,\mathbf{d}\left[\frac{1}{2}(6 - 2x - y)\right]\right)$

Out[120]= $\frac{3\,dx \wedge dy}{2}$

Out[121]= $-3(-3 + x)\,y\,dx \wedge dy$

In[122]:=
```
SetEuclideanNSpace[2, {x, y}, "Form"]
FormIntegral[xyDomain, xyForm]
EvaluateFormIntegrals[%]
```

Out[123]= $\int_{\mathcal{Domain}} -3(-3 + x)\,y\,dx \wedge dy$

Out[124]= $\frac{243}{2}$

See Also

FormIntegral · **PullbackForms** · **BasisPushforward** · **Integrate** · **Inactive** · **Activate**

Related Guides

- Calculus
- Derivatives

©1988–2015 Wolfram Research, Inc. All rights reserved. http://reference.wolfram.com/language